****************************************************************
/*      File name: 2022-01-03 Code for Creating the Misc Docket dataset	*/
/*		Purpose: Creating final document dataset of decisions from misc dockets 	*/
/*		Date: Branch created 2021-08-05 */
/*		Author: Jason Rantanen									*/
/*		Input files:											*/
/*			2021-12-31 CAFC Dockets.dta */
/*			appeals 2021-12-31.dta	*/
/*			2012.07.24 Mandamus Merits Dispositions by Federal Circuit (with appeal numbers).dta */
/*			2021-05-05 Export for defining set of decisions for review.dta */
/*			2021-06-09 PG origin key file.dta	*/
/*			2021-06-11 new orders to review.dta */
/*			2021-08-05 new coding for PG documents.dta */
/*			2021.08.06 Stage2_Dataset (updated, with PRG coding).dta */
/*			2022-01-03 JAR Additional Review (termination type).dta  */
/*		Output files: 	*/
/*			Misc Dockets Dataset - Dockets (final)	*/
/*			Misc Dockets Dataset - Decisions (final)	*/
/*			CAFC_available.doc	*/
****************************************************************

/*
The purpose of this code is to prepare and combine the foundational datasets (CAFC Dockets.dta, appeals.dta, and the 2012 PG study), add the additional coding that was done for this project, generate docket-level statistics for the methodology section, and output the final document-level dataset for analysis in the Results sections.

Note that this uses the 2021-12-31 version of the document dataset, which includes terminating documents in Misc Dockets through December 31, 2021.  

*/

ssc install asdoc // To make pretty tables


***********************************
/* Step 1: create a set of miscellaneous dockets from the full docket dataset */
cd "~/Dropbox/2021 Empirical Study of Mandamus at the CAFC/Data and Code/" 

use "2021-12-31 CAFC Dockets.dta", clear

split PACER_ID, gen(splitdocket) parse(-) destring
drop if splitdocket2 >999
drop (splitdocket1 splitdocket2)

save "Misc Dockets", replace

***********************************
/* Step 2: prepare the data from Paul Gugliuzza's 2012 study for merging with the data collected in 2021 */

use "2012.07.24 Mandamus Merits Dispositions by Federal Circuit (with appeal numbers).dta", clear

gen Appeal_Dockets_Full=Appeal_Dockets // This variable to be able to continue to track this block of information.

// Data are reshaped long with respect to appeal dockets.  
split Appeal_Dockets, generate(Short) parse(;) 
reshape long Short, i(Caption Citation) j(number)
drop Appeal_Dockets
gen wanted=strtrim(Short)
drop Short
rename wanted Short
drop if Short=="" & number!=1 
rename Short Appeal_Dockets
sort Appeal_Dockets
drop if Appeal_Dockets==""
//Data is now long with respect to both appeal docket numbers and documents.

//We are only keeping records from the 2012 dataset with an appeal docket number <1000.
split Appeal_Dockets, gen(split_docket) parse(-) destring
drop if split_docket2 >999
drop (split_docket1 split_docket2)

duplicates list Appeal_Dockets //There are 3 duplicates with respect to appeal docket.  These are instances in which there is more than one document associated with a single appeal docket. Four of these involve an unusual situation in which the petition is denied as to two dockets in the first order, then a later order denies a request for reconsideration and a petition in a new docket.  For these, the earlier document is kept as to the first two dockets (778 and 784) and the later document with respect to the third docket (790). The remaining pair of duplicates is because there was a limited hearing motion granted that again denied the petition on a slightly different ground.  Of these, the second one is kept.  

drop if Citation=="159 Fed.Appx. 992"
drop if Citation=="140 Fed.Appx. 256" & Appeal_Dockets=="2005-00778"
drop if Citation=="140 Fed.Appx. 256" & Appeal_Dockets=="2005-00784"

duplicates report Appeal_Dockets //At this point there should be no duplicates with respect to appeal docket number.
duplicates report Citation Caption //There are duplicates with respect to the document level; these are instances in which there is more than one appeal docket number associated wtih a document. 

rename * PG_= //data from the 2012 has a PG_ prefix.
rename PG_Appeal_Dockets Appeal_Dockets

//The lower court formatting needs to match the formatting in the other datasets.  This is addressed with a key file.
merge m:1 PG_LowerCourt using "2021-06-09 PG origin key file.dta"
drop if _merge==2
drop _merge

//Twelve dockets from the PG dataset do not have an equivalent in the key file.  

//Add the new coding that we added in connection with this project

merge m:1 PG_Citation using "2021-08-05 new coding for PG documents.dta"

drop _merge

save "PG Documents - prepared", replace


***********************************
/* Step 3: prepare the dataset from the Compendium's full document set. */

use "appeals 2021-12-31.dta", clear

drop if duplicate=="Yes"
drop if docType=="Errata"
drop if Misc_Docket_Order_Type=="Order – other" //These are non-dispositive orders, such as orders on motions to extend time.

//To make the script run faster, only use a limited set of fields to identify documents with misc dockets
keep uniqueID Appeal_Dockets

split Appeal_Dockets, generate(Short) parse(;) 
reshape long Short, i(uniqueID) j(number)
drop Appeal_Dockets
gen wanted=strtrim(Short)
drop Short
rename wanted Short
drop if Short=="" & number!=1 
rename Short Appeal_Dockets
sort Appeal_Dockets
drop if Appeal_Dockets==""

//Drop documents that are not associated with a miscellaneous docket number
split Appeal_Dockets, gen(split_docket) parse(-) destring
drop if split_docket2 >999
drop (split_docket1 split_docket2)
drop number

//Deduplicate to uniqueID uniqueness. 

duplicates drop uniqueID, force

//Add back in the fields from the main database
merge 1:1 uniqueID using "appeals 2021-12-31.dta", update replace

drop if _merge==2
drop _merge

gen Appeal_Dockets_Full=Appeal_Dockets // This variable to be able to continue to track this block of information.
gen CAFC_website=1 if DataSource=="CAFC Website"
replace CAFC_website=0 if CAFC_website==.

sort Appeal_Dockets

//Make documents long with respect to appeal dockets and uniqueID
split Appeal_Dockets, generate(Short) parse(;) 
reshape long Short, i(uniqueID) j(number)
drop Appeal_Dockets
gen wanted=strtrim(Short)
drop Short
rename wanted Short
drop if Short=="" & number!=1 
rename Short Appeal_Dockets
sort Appeal_Dockets
drop if Appeal_Dockets==""

duplicates report Appeal_Dockets uniqueID //Data is now unique with respect to appeal docket number + document level. 

duplicates report Appeal_Dockets // At this point each record is unique with respect to document and Appeal_Docket. 

// There are two duplicates with respect to appeal number.  These are instances in which there are multiple documents associated with a single docket number.  One of these is an order that was initially issued as nonprecedential (11428) and then made precedential (11544). Another is an order to show cause why sanctions should not be imposed (21270) followed by a sanctions order (21189). 

drop if uniqueID==11428 | uniqueID==21270

duplicates report Appeal_Dockets //There are no duplicates with respect to appeal docket at this stage. This means that every appeal docket is associated with one and only one terminating docket. 

save "Misc Appeal Documents temp.dta", replace

***********************************
/* Step 4: Combine 2012 dataset with documents from Compendium. */

merge 1:1 Appeal_Dockets using "PG Documents - prepared" // This merges the 2012 dataset with the document dataset from the Compendium. 

drop _merge


/* The next step is to update the data from Compendium with the data from the PG dataset */

//Encode precedential status to make it easier to work with
sort PrecedentialStatus
tab PrecedentialStatus
encode PrecedentialStatus, gen(PrecedentialStatus_n)
recode PrecedentialStatus_n (1=0) (2=1) (3=2)
label define PrecedentialStatus 0 Nonprecedential 1 Precedential 2 "PG_Published"
label values PrecedentialStatus_n PrecedentialStatus
tab PrecedentialStatus_n
drop PrecedentialStatus
rename PrecedentialStatus_n PrecedentialStatus


sort PG_Published
tab PG_Published
encode PG_Published, gen(PG_Published_n)
recode PG_Published_n (1=0) (2=1)
label define PG_Published 0 Nonprecedential 1 Precedential
label values PG_Published_n PG_Published
tab PG_Published_n
drop PG_Published
rename PG_Published_n PG_Published

//replace blank data with PG data
replace caseName= PG_Caption if caseName==""
replace Full_Cite= PG_Citation if Full_Cite==""
replace docDate=PG_Date if docDate==.
replace year=year(docDate) if year==.
replace PrecedentialStatus= PG_Published if PrecedentialStatus==.
replace PG_Author="Per Curiam" if PG_Author=="PC"
replace opinion1Author=PG_Author if opinion1Author==""
replace judge1= PG_Panel1 if judge1==""
replace judge2= PG_Panel2 if judge2==""
replace judge3= PG_Panel3 if judge3==""
replace Misc_Docket_Petition_Outcome= PG_Disposition if Misc_Docket_Petition_Outcome==""
replace Misc_Docket_Petition_Outcome="Dismissed" if Misc_Docket_Petition_Outcome=="Dismissed for lack of jurisdiction"

//For ease of working with the data, order type is encoded.
tab Misc_Docket_Order_Type
replace Misc_Docket_Order_Type="Dismissal order" if Misc_Docket_Order_Type=="Dimissed"
replace Misc_Docket_Order_Type="Dismissal order" if Misc_Docket_Order_Type=="Dismissed"
tab Misc_Docket_Order_Type
sort Misc_Docket_Order_Type

encode(Misc_Docket_Order_Type), gen(Misc_Docket_Order_Type_n)
recode Misc_Docket_Order_Type_n (2=1) (1=2)
label define Misc_Docket_Order_Type 1 "Merits Order" 2 "Dismissal Order" 3 "Order-transfer" 4 "Order-treat as appeal"
label values Misc_Docket_Order_Type_n Misc_Docket_Order_Type
tab Misc_Docket_Order_Type_n
drop Misc_Docket_Order_Type
rename Misc_Docket_Order_Type_n Misc_Docket_Order_Type

replace Appeal_Dockets_Full= PG_Appeal_Dockets_Full if Appeal_Dockets_Full==""

gen Dataset="Compendium" if uniqueID!=.
replace Dataset="PG 2012 only" if uniqueID==.
replace Dataset="Both" if uniqueID!=. & PG_Citation!=""

//All of the decisions added by the PG 2012 dataset (i.e.: that aren't in the Compendium) are unanimous decisions on petitions for writs of mandamus.
replace opinion1="Unanimous" if Dataset=="PG 2012 only"
replace Concurrence="No" if Dataset=="PG 2012 only"
replace Dissent="No" if Dataset=="PG 2012 only"
replace Misc_Docket_Appeal_Type="Writ of Mandamus" if Dataset=="PG 2012 only"


merge m:1 uniqueID Full_Cite using "2022-01-03 JAR Additional Review (termination type).dta", gen(merge5) update //This is coding of why petitions were dismissed.

drop if merge5==2
drop merge5

gen Terminating_Document = "Yes"

save "Combined Misc Document Dataset - temp", replace

//At this point we have a combined dataset with most fields populated with data from the 2012 study.  Fields that have not yet been transferred over are: PG_IssueReviewed and PG_Notes. 

use "Misc Dockets", clear
merge 1:1 Appeal_Dockets using "Combined Misc Document Dataset - temp", update force gen(merge2)



//There are a some documents from the PG dataset that do not have a corresponding match in the appeal dockets dataset.  These are from 2000 and the misc dockets from 2000 are not in the Compendium docket dataset.  These are dropped from the analysis.  We also want to drop dockets that are not Misc. Dockets.  In addition, there are some new documents that don't have a correponding number in the appeal dockets dataset.  We do not want to drop these.  

tab Appeal_Dockets if Year_Appeal_Filed==.
drop if Year_Appeal_Filed==. & Dataset=="PG 2012 only"

split Appeal_Dockets, parse(-) gen(temp)
destring(temp2), replace
drop if temp2>999
drop temp*

drop merge2

tab Year_Appeal_Filed

asdoc tab Year_Appeal_Filed, label font(Times New Roman) fs(12) save(Appeal_dockets_year.doc) replace title(\i) 

//This dataset will only contain data from 2008-March 2021.
drop if Year_Appeal_Filed<2008

//At this point the dataset consists of all appeal docket numbers at the Federal Circuit from 2000-March 2021 with a docket number <1000.  

replace Terminating_Document="No" if Terminating_Document==""

//As of 2021-06-09, almost all dockets between 2008 and 2020 have a terminating document.  Most ones that don't are special cases.  There are some terminating documents mising for 2020 and 2021 that need to be flagged or collected.

tab Year_Appeal_Filed Terminating_Document, missing

tab Appeal_Type_fm_Docket Appeal_Outcome_frm_Docket if Year_Appeal_Filed == 2008 & Terminating_Document=="No"
tab Appeal_Dockets Appeal_Type_fm_Docket if Year_Appeal_Filed == 2008 & Terminating_Document=="No"
tab Appeal_Dockets Appeal_Outcome_frm_Docket if Year_Appeal_Filed == 2008 & Terminating_Document=="No" & Appeal_Type_fm_Docket=="Writ of Mandamus"
//Of the 7 dockets from 2008 that are missing a terminating document, 5 are writs of mandamus (the other 2 are petitions for permission to appeal). Of the petitions for a writ of mandamus, 3 were dismissed, 1 was granted, and 1 had an outcome of "Other." The one that was granted is 2009-00892. 

tab Appeal_Type_fm_Docket Appeal_Outcome_frm_Docket if Year_Appeal_Filed > 2008 & Year_Appeal_Filed < 2020 & Terminating_Document=="No"
tab Appeal_Dockets Appeal_Type_fm_Docket if Year_Appeal_Filed > 2008 & Year_Appeal_Filed < 2020 & Terminating_Document=="No"
//Of the dockets that do not have a terminating document for 2009-2019, all but 2 are in the "other" classification. The two that are missing are 2009-00905 and 2010-00950.  Both were denied. Electronic copies of these documents are no longer available.

//Dropping unused fields
drop PACER_Gen district_id USPTO_case_row_wide Orig_Trib_PACER Misc_Docket case_number_wide Last_Activity_Date Docket_fileName number origin duplicate url n_Patents_in_Suit UtilityPatent DesignPatent PlantPatent Issue_Utility Issue_PSM Issue_102 Issue_103 Issue_Enablement Issue_WD Issue_Definiteness Patents_in_Suit Issue_ObvTypDblPatenting Issue_ClaimConstruction NewFileName 

//Testing completeness of other fields

tab Year_Appeal_Filed if TribOfOrigin =="", missing
tab Year_Appeal_Filed Appeal_Outcome_frm_Docket if Terminating_Document == "Yes" & judge1 ==""
tab Year_Appeal_Filed Appeal_Outcome_frm_Docket if Terminating_Document == "Yes" & opinion1Author ==""
tab Year_Appeal_Filed Appeal_Outcome_frm_Docket if Terminating_Document == "Yes" & opinion1 ==""
tab Year_Appeal_Filed Appeal_Outcome_frm_Docket if Terminating_Document == "Yes" & Misc_Docket_Appeal_Type ==""
tab Year_Appeal_Filed Appeal_Outcome_frm_Docket if Terminating_Document == "Yes" & Misc_Docket_Order_Type ==.
tab Year_Appeal_Filed Appeal_Outcome_frm_Docket if Terminating_Document == "Yes" & Misc_Docket_Petition_Outcome ==""
tab Year_Appeal_Filed Appeal_Outcome_frm_Docket if Terminating_Document == "Yes" & Dissent ==""
tab Year_Appeal_Filed Appeal_Outcome_frm_Docket if Terminating_Document == "Yes" & Concurrence ==""
tab Year_Appeal_Filed Appeal_Outcome_frm_Docket if Terminating_Document == "Yes" & opinion1 ==""
tab Year_Appeal_Filed Appeal_Outcome_frm_Docket if Terminating_Document == "Yes" & opinion1Author ==""

merge 1:1 PACER_ID using "2021-05-07 Export for defining set of decisions for review.dta" //This provides information on whether the docket numbers where included in the first pass review done by JJA and PG.

drop _merge

replace CAFC_website=0 if CAFC_website==. 

//These are records that only appear in the docket dataset and we were unable to obtain the actual terminating document at all. 

save "Misc Dockets Dataset - Dockets (final)", replace

******
/* The Dataset in Docket form is now set */
******


***********************************
/* Step 5: Generate statistics for methodology section of article. */

//The next step is to run the statistics on the docket-level data

//Date for graph of Misc Dockets by year

asdoc tab Year_Appeal_Filed, label font(Times New Roman) fs(12) save(CAFC_available.doc) replace title(Misc Dockets by year appeal filed) 

//Analysis of whether documents are available on CAFC website: 

label define CAFC_website 1 Yes 0 No, replace
label values CAFC_website CAFC_website
label var CAFC_website "Document available on CAFC website"

tab Year_Appeal_Filed CAFC_website, missing
asdoc tab Year_Appeal_Filed CAFC_website, missing label font(Times New Roman) fs(12) save(CAFC_available.doc) append title(Terminating document on website by year appeal filed) 

label define DistrictCourt 1 "from district court" 0 "from other origin", replace
label values DistrictCourt DistrictCourt

label var Terminating_Document "Terminating document in dataset"

tab Year_Appeal_Filed Terminating_Document, missing
asdoc tab Year_Appeal_Filed Terminating_Document, label font(Times New Roman) fs(10) save(CAFC_available.doc) append title(Outcomes of Petitions for Writ of Mandamus) 

//Additional statistics on docket-level data

tab year Appeal_Type_fm_Docket

tab Appeal_Outcome_frm_Docket DistrictCourt if Appeal_Type_fm_Docket=="Writ of Mandamus"

//Data for beginning of Results sections

asdoc tab Year_Appeal_Filed Appeal_Type_fm_Docket, label font(Times New Roman) fs(12) save(CAFC_available.doc) append title(Appeal type by year filed - docket) 

asdoc tab origin_PACER Appeal_Type_fm_Docket if Appeal_Type_fm_Docket!="Other", label font(Times New Roman) fs(12) save(CAFC_available.doc) append title(Appeal type by year filed - docket) 

tab TribOfOrigin if Year_Appeal_Filed==2021 & Appeal_Type_fm_Docket=="Writ of Mandamus"

//Time to resolution

gen Time_To_Decision=docDate - DateFiledinCOA if docDate!=.
replace Time_To_Decision=. if Time_To_Decision<0
tab Time_To_Decision Misc_Docket_Order_Type

save "Misc Dockets Dataset - Dockets (final)", replace


***********************************
/* Step 6: Generate document level dataset */


//Remember, "uniqueID" is the unique identifer for each document, except for the documents from the PG dataset (for these, the unique identifier is PG_Citation). PACER_id or Appeal_Docket is the unique identifier for each docket. 

drop if Terminating_Document=="No"

duplicates report uniqueID PG_Citation docDate year caseName Full_Cite docType enBanc judge1 judge2 judge3 opinion1 opinion1Author opinion2 opinion2Author opinion3 opinion3Author DisputeType DispGeneral FileName Replaced Replaced_Notes Orig_Trib_Dockets Appellant_Type_Primary Patent_Owner_Win Appellant_Type_Cross CloudLink Misc_Docket_Appeal_Type Misc_Docket_Petition_Outcome Notes_DPK Dissent Concurrence Appeal_Dockets_Full Misc_Docket_Order_Type PG_Caption PG_Citation PG_number PG_Date PG_DocketNumber PG_IssueReviewed PG_LowerCourt PG_Disposition PG_Author PG_Panel1 PG_Panel2 PG_Panel3 PG_Notes PG_Appeal_Dockets_Full PrecedentialStatus PG_Published Dataset PG_Citation Appeal_Type_fm_Docket Appeal_Outcome_frm_Docket DistrictCourt TribOfOrigin

//This indicates that there are a moderate number of duplicate documents in the dataset.  However, the additional duplicates reports indicate that all of these documents are identical with respect to all relevant document and docket-level fields other than Appeal_Docket number. In addition, the Appeal_Dockets_Full field retains the full list of dockets should it be necessary.

sort uniqueID DateFiledinCOA // This will retain the earliest docket filed in the COA.

duplicates drop uniqueID PG_Citation, force // This drops all duplicate copies of a document.  The effect of this is to eliminate the extra dockets while retaining the data at the document level. In addition, the Appeal_Dockets data is preserved at the Appeal_Dockets_Full level.  Consequently, we no longer need the PACER_id field or the temporary Appeal_Dockets fields. 

merge 1:1 uniqueID PG_Citation using "2021-06-11 new orders to review.dta", update force //This is a determination of whether to include in the Stage 2 analysis additional documents identified during final dataset construction.

drop _merge

merge 1:1 uniqueID Full_Cite using "/Users/jrantanen/Dropbox/2021 Empirical Study of Mandamus at the CAFC/Data and Code/2022-01-03 Include in Stage 2 Review.dta", update // This is the final determination of whether to include documents in the Stage 2 review.

drop _merge

//Dropping fields that we are not using in the final analysis

drop OriginatingCase Docketavailable Notes notes DisputeType DispGeneral FileName Replaced Replaced_Notes  Orig_Trib_Dockets Appellant_Type_Primary Appellant_Type_Cross Patent_Owner_Win Notes_DPK Notes2 PG_Caption PG_Date PG_DocketNumber PG_IssueReviewed PG_LowerCourt PG_Disposition PG_Author PG_Panel1 PG_Panel2 PG_Panel3 PG_Notes PG_Appeal_Dockets_Full temp1 missing Appeal_Dockets appealNumber PG_number PG_Published Terminating_Document PG_Citation Include_In_DCT_Coding_First_Pass JARClassification PACER_Appeal_Docket CoderBKNotes DispText uniqueID_relatedDecision

order uniqueID docDate year caseName Full_Cite docType enBanc judge1 judge2 judge3 opinion1 opinion1Author opinion2 opinion2Author opinion3 opinion3Author CloudLink, before(Docket_Title)



order DistrictCourt, before(judge1)
order TribOfOrigin, before(DistrictCourt)
order Concurrence Appeal_Dockets_Full, before(CloudLink)
order Dissent , before(Appeal_Dockets_Full)
order Misc_Docket_Appeal_Type Misc_Docket_Appeal_Type Misc_Docket_Order_Type Misc_Docket_Petition_Outcome, before(judge1)
order PrecedentialStatus, before(enBanc)

label var docDate "Date of CAFC document"
label var year "Year of CAFC document"

label var Misc_Docket_Appeal_Type "Type of Misc Docket (from order)"
label var Misc_Docket_Petition_Outcome "Outcome of order (from order)"
label var Misc_Docket_Order_Type "Type of order (from order)"

label var Issue1 "RA-coded Issue 1"
label var Issue2 "RA-coded Issue 2"
label var Issue3 "RA-coded Issue 3"
label var Dataset "source of record"

replace enBanc="No" if Dataset=="PG 2012 only" //None of the documents added from the PG 2012 dataset were en banc

replace Misc_Docket_Petition_Outcome="Transferred" if Misc_Docket_Petition_Outcome=="Transfered"

merge 1:1 uniqueID Full_Cite using "2021.08.06 Stage2_Dataset (updated, with PRG coding).dta" //This is the coding from PG and JJA.

gen Stage2_Dataset = 1 if _merge==3
replace Stage2_Dataset = 0 if Stage2_Dataset==.

drop _merge


gen PetitionLink="https://storage.googleapis.com/compendium_testing/"+Petition_FileName if Petition_FileName!=""
replace Docket_cloudLink="https://storage.googleapis.com/compendium_testing/"+Docket_Filename if Docket_cloudLink=="" & Docket_Filename!=""
order PetitionLink, before(CaptionFilename)

merge 1:1 uniqueID Full_Cite using "/Users/jrantanen/Dropbox/2021 Empirical Study of Mandamus at the CAFC/Data and Code/2021-12-31 Subject Matter Coding.dta" 
drop if _merge==2
drop _merge

*** Document level statistics for methodology discussion of article ***

tab Misc_Docket_Petition_Outcome CAFC_website if Misc_Docket_Appeal_Type=="Writ of Mandamus"
tab Misc_Docket_Order_Type CAFC_website if Misc_Docket_Appeal_Type=="Writ of Mandamus"
tab ReasonAppealTermination CAFC_website if Misc_Docket_Appeal_Type=="Writ of Mandamus"
tab year CAFC_website if Misc_Docket_Appeal_Type=="Writ of Mandamus"

save "Final Document dataset with all fields.dta", replace //This copy of the dataset has all fields in case it is necessary to pull in this data.



drop Docket_Title Appeal_Outcome_frm_Docket CaptionFilename CAFC_website Issue1 Issue2 Issue3 Dataset JJA_Action_Requested PG_Action_Requested JJA_Stage_2_Coder_Notes PG_Stage_2_Coder_Notes //This is to make the dataset more manageable. 

//Delete temporary files
erase "Misc Dockets.dta"
erase "PG Documents - prepared.dta"
erase "Misc Appeal Documents temp.dta"
erase "Combined Misc Document Dataset - temp.dta"



******
/* The Dataset in Document form is now set */
******
save "Misc Dockets Dataset - Decisions (final)", replace
